program define fx1_byff
   version 15.1
   syntax varlist [if] [pweight], t1(varname) [OPTS(str) ALLFIVE]
	marksample touse
	if "`weight'" ~= "" {
	   local wt [`weight'`exp']
   }	
	tempname fx0 fx1
   forvalues f = 0/1 {
	   di _n(2) "** FF Pref `f' **"
      fx1 `varlist' if `touse' & ff2==`f' `wt', t1(`t1') `allfive' nodraw
      mat `fx`f'' = r(fx)
   }
	
	* combine plots horizontally
	local gopts scheme(s1mono) plotregion(style(none)) mlabel format(%3.0f)
   coefplot ///
	   (matrix(`fx1'[1,]), ci("`fx1'[5,] `fx1'[6,]") msymbol(O) mlabpos(10)), bylabel(Favor New Laws) ///
		||  ///
	   (matrix(`fx0'[1,]), ci("`fx0'[5,] `fx0'[6,]") msymbol(O) mlabpos(10)), bylabel(Oppose New Laws) ///	   
		coeflabel(none = "No Action" low = "Cut 5%" high = "Cut 25%" /// cover all possible labels
		   c0="No Action" c1="Cut 5%, Cost 10%" c2="Cut 5%, Cost 4%" ///
         c3="Cut 25%, Cost 10%" c4="Cut 25%, Cost 4%") ///
		byopts(rows(1) legend(off)) subtitle(, fcolor(white)) `gopts' `opts'
end
